```
In[1]:= ClearAll["Global`*"]

In[2]:= k = 4

Out[2]= 4
```

This code attempts computing analytically the expectations for the various terms in Eq. 28 (Sec B1) of the Supplementary Information. The code computes the expectations for n = {2,3,...k} (with k small enough), and uses these values to verify the equations in Sec B1. To verify the equations for larger values of n, increase the value k above (the computing time grows exponentially; the values have been verified for n = 10)

Outer product

```
In[3]:= Op[a_, b_] := Outer[Times, a, b]
```

Build matrix with arbitrary coefficients and zero on diagonal

```
In[4]:= BuildB[n_] := Module[{A, B}, A = Table[b[i, j], {i, 1, n}, {j, 1, n}];
          B = A - DiagonalMatrix[Diagonal[A]];
          B]
```

Remove the mean from matrix B

```
In[5]:= BuildC[n_] := Module[{B, ones, m, K},
          B = BuildB[n];
          ones = Table[1, n];
          m = 1/n Flatten[Transpose[B].ones]; K = B - Op[ones, m]; K]
```

This is the right hand side of the equations

```
In[6]:= Buildb[n_] := Table[1/α, n]
```

Build the initial guess

```
In[7]:= Buildy0[n_] := Table[1/α, n] - 1/α^2 BuildC[n].Table[1, n]
```

This is the matrix M (which depends on the correlation; this is the version for matrices with independent entries, as specified in Sec B1)

```
In[8]:= BuildM[n_] := Module[{M}, M = IdentityMatrix[n] + 1/α BuildC[n]; M]
```

These are the expectations for the powers of Bij

```
In[9]:= r6 := Flatten[Table[b[i, j]^6 → μ6, {i, 1, n}, {j, 1, n}]]

In[10]:= r5 := Flatten[Table[b[i, j]^5 → μ5, {i, 1, n}, {j, 1, n}]]

In[11]:= r4 := Flatten[Table[b[i, j]^4 → μ4, {i, 1, n}, {j, 1, n}]]

In[12]:= r3 := Flatten[Table[b[i, j]^3 → μ3, {i, 1, n}, {j, 1, n}]]
```

In[13]:=
```
r2 := Flatten[Table[b[i, j]^2 → μ2, {i, 1, n}, {j, 1, n}]]
```

In[14]:=
```
rrem := Flatten[Table[b[i, j] → 0, {i, 1, n}, {j, 1, n}]]
```

Compute expectation given a matrix of expressions

In[15]:=
```
ExpandMat[Z_] := Total[Total[ExpandAll[Z] /. r6 /. r5 /. r4 /. r3 /. r3 /. r2 /. r2 /. r2 /. rrem]]
```

In[16]:=
```
ExpandExpression[x_] := ExpandAll[x] /. r6 /. r5 /. r4 /. r3 /. r3 /. r2 /. r2 /. r2 /. rrem
```

Having set up the problem, we compute all the terms in Eq 28

1. Term $b^T b$. First, we compute the expectation for this term for different values of n

In[17]:=
```
btb = FullSimplify[Table[Buildb[n].Buildb[n], {n, 2, k}]]
```

Out[17]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}, \frac{4}{\alpha^2}\right\}$$

The formula in Sec B1 reads

In[18]:=
```
predictedbtb = Table[n / α ^ 2, {n, 2, k}]
```

Out[18]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}, \frac{4}{\alpha^2}\right\}$$

Make sure that the terms match

In[19]:=
```
FullSimplify[btb - predictedbtb]
```

Out[19]=

```
{0, 0, 0}
```

2. Term $b^T y_0$. First, we compute the expectation for this term for different values of n

In[20]:=
```
bty0 = FullSimplify[Table[Buildb[n].Buildy0[n], {n, 2, k}]]
```

Out[20]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}, \frac{4}{\alpha^2}\right\}$$

The formula in Sec B1 reads

In[21]:=
```
predictedbty0 = Table[n / α ^ 2, {n, 2, k}]
```

Out[21]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}, \frac{4}{\alpha^2}\right\}$$

Make sure that the terms match

In[22]:=
```
FullSimplify[bty0 - predictedbty0]
```

Out[22]=

```
{0, 0, 0}
```

3. Term $y_0^T\, y_0$. First, we compute the expectation for this term for different values of n

In[23]:=
```
y0ty0 = FullSimplify[Table[ExpandExpression[Buildy0[n].Buildy0[n]], {n, 2, k}]]
```

Out[23]=

$$\left\{\frac{2\,\alpha^2+\mu 2}{\alpha^4},\ \frac{3\,\alpha^2+4\,\mu 2}{\alpha^4},\ \frac{4\,\alpha^2+9\,\mu 2}{\alpha^4}\right\}$$

The formula in Sec B1 reads

In[24]:=
```
predictedy0ty0 = Table[n / α ^ 2 + 1 / α ^ 4 μ2 (n - 1) (n - 1), {n, 2, k}]
```

Out[24]=

$$\left\{\frac{2}{\alpha^2}+\frac{\mu 2}{\alpha^4},\ \frac{3}{\alpha^2}+\frac{4\,\mu 2}{\alpha^4},\ \frac{4}{\alpha^2}+\frac{9\,\mu 2}{\alpha^4}\right\}$$

Make sure that the terms match

In[25]:=
```
FullSimplify[y0ty0 - predictedy0ty0]
```

Out[25]=
```
{0, 0, 0}
```

4. Term $b^T\, M\, y_0$. First, we compute the expectation for this term for different values of n

In[26]:=
```
btMy0 = Table[ExpandExpression[FullSimplify[Buildb[n].BuildM[n].Buildy0[n]]], {n, 2, k}]
```

Out[26]=

$$\left\{\frac{2}{\alpha^2},\ \frac{3}{\alpha^2},\ \frac{4}{\alpha^2}\right\}$$

The formula in Sec B1 reads

In[35]:=
```
predictedbtMy0 = Table[n / α ^ 2, {n, 2, k}]
```

Out[35]=

$$\left\{\frac{2}{\alpha^2},\ \frac{3}{\alpha^2},\ \frac{4}{\alpha^2}\right\}$$

Make sure that the terms match

In[36]:=
```
FullSimplify[btMy0 - predictedbtMy0]
```

Out[36]=
```
{0, 0, 0}
```

5. Term $y_0^T\, M\, y_0$. First, we compute the expectation for this term for different values of n

In[29]:=
```
y0tMy0 = Table[ExpandExpression[FullSimplify[Buildy0[n].BuildM[n].Buildy0[n]]], {n, 2, k}]
```

Out[29]=

$$\left\{\frac{2}{\alpha^2}-\frac{\mu 3}{2\,\alpha^5},\ \frac{3}{\alpha^2}-\frac{4\,\mu 3}{3\,\alpha^5},\ \frac{4}{\alpha^2}-\frac{9\,\mu 3}{4\,\alpha^5}\right\}$$

The formula in Sec B1 reads

In[30]:=
```
predictedy0tMy0 = Table[(n) / α ^ 2 - (n - 1) ^ 2 / (n α ^ 5) μ3, {n, 2, k}]
```
Out[30]=

$$\left\{\frac{2}{\alpha^2} - \frac{\mu 3}{2\,\alpha^5},\ \frac{3}{\alpha^2} - \frac{4\,\mu 3}{3\,\alpha^5},\ \frac{4}{\alpha^2} - \frac{9\,\mu 3}{4\,\alpha^5}\right\}$$

Make sure that the terms match

In[31]:=
```
FullSimplify[y0tMy0 - predictedy0tMy0]
```
Out[31]=
```
{0, 0, 0}
```

6. Term $y_0^T M^T M y_0$. First, we compute the expectation for this term for different values of n

In[32]:=
```
y0tMtMy0 = Table[ExpandExpression[
    FullSimplify[Buildy0[n].Transpose[BuildM[n]].BuildM[n].Buildy0[n]]], {n, 2, k}]
```
Out[32]=

$$\left\{\frac{2}{\alpha^2} - \frac{\mu 2^2}{4\,\alpha^6} + \frac{\mu 4}{4\,\alpha^6},\ \frac{3}{\alpha^2} + \frac{14\,\mu 2^2}{3\,\alpha^6} + \frac{4\,\mu 4}{9\,\alpha^6},\ \frac{4}{\alpha^2} + \frac{321\,\mu 2^2}{16\,\alpha^6} + \frac{9\,\mu 4}{16\,\alpha^6}\right\}$$

The formula in Sec B1 reads

In[33]:=
```
predictedy0tMtMy0 =
  Table[n / α ^ 2 + μ4 (n - 1) ^ 2 / (n ^ 2 α ^ 6) + (-1 + n) (3 - 6 n + 4 n^2 - 3 n^3 + n^4) μ2^2 / (n ^ 2 α ^ 6), {n, 2, k}]
```
Out[33]=

$$\left\{\frac{2}{\alpha^2} - \frac{\mu 2^2}{4\,\alpha^6} + \frac{\mu 4}{4\,\alpha^6},\ \frac{3}{\alpha^2} + \frac{14\,\mu 2^2}{3\,\alpha^6} + \frac{4\,\mu 4}{9\,\alpha^6},\ \frac{4}{\alpha^2} + \frac{321\,\mu 2^2}{16\,\alpha^6} + \frac{9\,\mu 4}{16\,\alpha^6}\right\}$$

Make sure that the terms match

In[34]:=
```
FullSimplify[y0tMtMy0 - predictedy0tMtMy0]
```
Out[34]=
```
{0, 0, 0}
```